
Removing Duplicates Using SAS® 

We live in a world of data – small data, big data, and data in every conceivable size between small and big. In today’s 

world data finds its way into our lives wherever we are. We talk about data, create data, read data, transmit data, 

receive data, and save data constantly during any given hour in a day, and we still want and need more. So, we collect 

even more data at work, in meetings, at home, using our smartphones, in emails, in voice messages, sifting through 

financial reports, analyzing profits and losses, watching streaming videos, playing computer games, comparing sports 

teams and favorite players, and countless other ways. Data is growing and being collected at such astounding rates all 

in the hopes of being able to better understand the world around us. As SAS professionals, the world of data offers 

many new and exciting opportunities, but also presents a frightening realization that data sources may very well 

contain a host of integrity issues that need to be resolved first. 

 

An issue found in some data sets is the presence of duplicate rows and/or duplicate keys. When found, SAS can be 

used to remove any unwanted data. Note:  Before duplicates are removed, be sure to consult with your organization’s 

data analyst or subject matter expert to see if removal is necessary. It’s better to be safe than to be sorry. 

 

This tip illustrates two different methods to remove duplicate observations (or rows) from data sets (or tables) based on 

the row’s values and/or keys using SAS®. Each example is illustrated using a single data set, MOVIES. The Movies data set 

contains 26 rows, and has a structure consisting of six columns. Title, Category, Studio, and Rating are defined as 

character columns; and Length and Year are defined as numeric columns. The Movies data set contains two duplicate 

rows – Brave Heart and Rocky; and two duplicate Title keys – Forrest Gump and The Wizard of Oz, shown below. 

Method #1 – Using PROC SORT 

The first method, and one that is popular with SAS professionals everywhere, uses PROC SORT to remove duplicates. 

The SORT procedure supports three options for the removal of duplicates: NODUPRECS, NODUPKEYS, and DUPOUT=. 

 

The NODUPRECS (or NODUP) Option 

By specifying the NODUPRECS (or NODUPREC) (or NODUP) option with PROC SORT, rows with identical values for all 

columns are removed from the output data set. The resulting output data saw the removal of the duplicate rows for 

Brave Heart and Rocky because they have identical data for all columns. 

 
PROC SORT DATA=Movies 

           OUT=Movies_Sorted_NoDuprecs 

          NODUPRECS ; 

  BY Title ; 

RUN ; 

 
The NODUPKEYS (or NODUPKEY) Option 

By specifying the NODUPKEYS (or NODUPKEY) option with PROC SORT, rows with duplicate keys are automatically 

removed from the output data set. The resulting output data set saw the removal of all the duplicate rows for Brave 

Heart, Forrest Gump, Rocky and The Wizard of Oz because they have duplicate keys data for the column, Title. 

 
PROC SORT DATA=Movies 

           OUT=Movies_Sorted_NoDupkeys 

          NODUPKEYS ; 

  BY Title ; 

RUN ; 

 
The DUPOUT= Option 

A DUPOUT= option is specified with PROC SORT to identify duplicate rows before actually removing them from a data 

set. The DUPOUT= option is used with either the NODUPKEYS or NODUPRECS option to name a data set that will 

contain duplicate keys or duplicate rows. The DUPOUT= option is generally used when the data set is too large for 

visual inspection. In the next code example, the DUPOUT= and NODUPKEY options are specified. The resulting output 

data set contains the duplicate rows for Brave Heart, Forrest Gump, Rocky and The Wizard of Oz. 

 
PROC SORT DATA=Movies 

        DUPOUT=Movies_Sorted_Dupout_NoDupkey 

        NODUPKEY ; 

  BY Title ; 

RUN ; 

 
In the next example, the DUPOUT= and NODUPRECS options are specified. The resulting output data set contains the 

duplicate rows for Brave Heart and Rocky because these rows have identical data for all columns. 

 
PROC SORT DATA=Movies 

        DUPOUT=Movies_Sorted_Dupout_NoDuprecs 

        NODUPRECS ; 

  BY Title ; 

RUN ; 

 

Note:  Although the removal of duplicates using PROC SORT is popular with many SAS professionals, an element of 

care should be given to using this method when processing big data sets. Because sort operations are time consuming 

and CPU-intensive operations, requiring as much as three times the amount of space to sort a data set, excessive 

demand is placed on system resources. Instead, SAS professionals may want to consider using PROC SUMMARY with 

the CLASS statement to avoid the need for sorting altogether, see Method #2. 

Method #2 – Using PROC SUMMARY with the CLASS Statement 

The second method of removing duplicates uses PROC SUMMARY with the CLASS statement. Using PROC SUMMARY 

with the CLASS statement provides SAS professionals with a more efficient alternative than PROC SORT, and other 

methods, by avoiding the need for sorting in advance. Without the sorting requirement, considerably less system 

resources are needed to identify duplicates. But two additional aspects make this method effective: the specification 

of a CLASS statement to collapse rows with the same column values and the creation of a _FREQ_ column containing 

the number of occurrences. As shown in the example, a WHERE statement, WHERE= data set option, or SQL WHERE-

clause is specified to select rows with multiple occurrences (duplicates) with the WHERE-clause expression Dupkey > 1. 

 
PROC SUMMARY DATA=Mydata.Movies_dups2 NWAY ; 

  CLASS Title ; 

  OUTPUT OUT=Movies_Summary_NoDupkey(DROP=_type_) ; 

RUN ; 

PROC PRINT DATA=Movies_Summary_NoDupkey(RENAME=(_FREQ_ = Dupkey)) NOOBS ; 

  WHERE Dupkey > 1 ; 

RUN ; 

