
Abstract Description : This presentation shows the programming steps in the development
of an Exploratory Data Analysis (EDA) application.

Purpose : The case study consists of six files which summarize each variable in
any specified data set. The discipline shows how to write programs
and conduct both unit and integration tests. The emphasis is on writ-
ing small, easy to read programs, which may be converted to macros
at a later time. Global macro variables are used as parameters. This
unit is the first of four of a planned day-long seminar on applications
development for programmers.

Audience : beginning programmers and intermediate to advanced users

Applications Design and Development, continued SESUG 2015

Quote Information is the difference
that makes a difference

Gregory Bateson, 1904–1980
Steps to an Ecology of Mind, 1972

Checklist of Ideas • cardinality ratio n-levels, n-obs

• categories of variables few many row-id
discrete continuous unique

• guarantee quality, readable, tested

• list contains items, tasks, or lists Polish notation

• mnemonics DJD, HIPO, LATCH

• requirements, design, 80/20, specifications change notice

• style guide naming conventions

• data structure less than 1% of statements
99% of success

• global macro variables parameters

• hierarchy of program types module, routine, or subroutine

• know thy options prepare for testing

• small is beautiful easy to test

• use SAS naming conventions avoid quirks!

This table shows the time spent during applications development.time budget

Phase Time Action Time
design 1/2 understand problem: 1/3

education and research
development coding 1/6

testing 1/2 component or unit test 1/4
systems or integration test 1/4

The Mythical Man-Month, by Fred Brooks, on the development of the IBM 360

EDA requirements for each variable in a data set, perform frequency

module routine subroutine
01-data-class
proc-1-freq make-list-vars

proc-list-vars proc-freq

Compare to Change Notice requirements on page 14.

Applications Design and Development, continued SESUG 2015

discipline of
applications
development

• hard-coded absolute

• soft-coded relative
use macro variables as parameters

• split into subroutine.sas
subroutine-test.sas

• testing unit
integration

Research, Contents Procedure

contents
sashelp.class

1 PROC contents data = sashelp.class

Data Set Name SASHELP.CLASS Observations 19
Member Type DATA Variables 5

Created Tuesday, May 24, 2011 01:52:28 PM [granularity]
Label Student Data

Folder ...\core\sashelp\ [libref]
Filename class.sas7bdat [data set name]

Location Alphabet Time Category Hierarchy

5 Ways of Organizing Information

Alphabetic List of Variables and Attributes

Var
Num Variable Type Len

3 Age Num 8
4 Height Num 8
1 Name Char 8
2 Sex Char 1
5 Weight Num 8

Location Alphabet Time Category Hierarchy

The objective of list processing is to have SAS write statements like this program.desired program

%let in_data = sashelp.class;

proc freq data = &in_data ...
tables Age ...
tables Height ...
tables Name ...
tables Sex ...
tables Weight ...

proc display data = out_frequencies;

slide 10

The output data sets are concatenated and look like this.desired output

valu valu
name char num COUNT PERCENT
----- ------ ---- ----- -------
Name Alfred . 1 5.263

Alice . 1 5.263
...
Sex F . 9 47.368

M . 10 52.631
Age 11.0 2 10.526

12.0 5 26.315
...
Height 51.3 1 5.263

56.3 1 5.263
slide 11

Bottom Up, Frequency Procedure

This program is the first step in applications development, getting an example program in SAS.step.1 hard-coded
Concatenation requires that the output data sets have a consistent data structure, i.e., same
variable names. This program shows the names of variables in an output data set from the
frequency procedure.

1 proc freq data = sashelp.class;
2 tables sex / out = out_freq_sex;
3 tables age / out = out_freq_age;
4 proc sql; describe table out_freq_sex;
5 describe table out_freq_age;
6 quit;

r11-freq.log

create table out_freq_sex
Sex char(1),
count num label = Frequency Count,
percent num label = Percent of Total Frequency

create table out_freq_age
Age num,

slide 12

Notes: 1. Discipline step.1: hard-coded statements. 2. describe table writes to the
log. 3. file-naming conventions: r1?-freq for frequency procedure.

This program shows the allocation of two macro variables, in data and var, as the param-step.2 soft-coded
eters for a sub-program.

1 %let in_data = sashelp.class;
2 %let var = sex;
3

4 proc freq data = &in_data;
5 tables &var / out = out_freq;
6

7 proc sql; describe table &syslast;
8 quit;

slide 13

Notes: 1. Discipline step.2: soft-coded statements, using global macro variables as parame-
ters. 2. syslast is a global automatic macro variable, which contains the two-level name
of the data set created in the previous step.

This sub-program shows the references of the two macro variables, in data and var, as thestep.3.1 split into
called parameters for this called sub-routine.

1 proc freq data = &in_data;
2 tables &var / noprint
3 out = out_freq
4 (rename = (&var = valu_&type ));
5

6 proc sql; describe table &syslast;
7 quit;
8

9 proc append data = &syslast
10 base = out_data;

slide 14

Notes: 1. Discipline step.3: split soft-coded program into two programs. 2. Lines 5-6, see
program describe, page 11. 3. Style guide: use white space to group keywords and
arguments in vertical columns.

This program shows the allocation of two macro variables, in data and var, and the call ofstep.3.2 . . . and caller
the sub-routine.

1 %let in_data = sashelp.class;
2

3 %let var = sex;
4 %let type = c;
5 %include ’r13-freq.sas’/source2;
6

7 %let var = age;
8 %let type = n;
9 %include ’r13-freq.sas’/source2;

10

11 proc print data = &syslast;
12 run;

slide 15

Notes: 1. lines 5, 9; option source2 is a global option, shown in next program, and option
of the %include statement; it echoes the statements in the file to the log. 2. The object
is described first and then the verb. This is an example of a task written in Reverse Polish
Notation, i.e., the verb after the object comes. Why Yoda famous was. 3. Polish notation: 2+2=4 ::
infix notation; + 2 2 :: prefix notation; 2 2 + reverse notation.

debugging
NOTE: Appending WORK.OUT_FREQ to WORK.OUT_DATA.
WARNING: Variable valu_N was not found on BASE file.

ˆ
WARNING: Variable valu_C was not found on DATA file.

ˆ
ERROR: No appending done because of anomalies listed above.
NOTE: Statements not processed because of errors noted above.

slide 16

Data Structure

Data sets to be appended must have same variable names. The variable is renamed in thepolishing data
structure frequency procedure output data set and the data step then adds the other variable name.

1 PROC freq data = &in_data;
2 tables &var / noprint
3 out = out_freq
4 (rename =(&var = valu_&type) );
5

6 DATA freq_values;
7 attrib valu_c length = $32
8 valu_n length = 8;
9 set &syslast;

10

11 PROC append data = &syslast
12 base = out_data;
13 run;

slide 17

Notes: 1. Data structure may be as small as <1% of program statements. 2. Style guide,
the naming convention for variables valu-c and valu-n depends on the parameter type.

This program shows the principle of allocating a macro variable – proc name — to replacepolishing test program
any set of two or more occurences of the same string.

1 options source2;
2 %let in_data = sashelp.class;
3 %let proc_name = ’r14-freq.sas’;
4

5 %let var = sex;
6 %let type = c;
7 %include &proc_name;
8

9 %let var = age;
10 %let type = n;
11 %include &proc_name;
12

13 proc print data = &syslast;
14 run;

slide 18

Notes: 1. Option source2 is a global option; it has been moved from the %include state-
ment. 2. Macro variable proc-name reduces typing and the need for find-and-replace.
Compare to previous program, r13-freq-test.

output, first draft Obs valu_c valu_n COUNT PERCENT

1 F . 9 47.3684
2 M . 10 52.6316
3 11 2 10.5263
4 12 5 26.3158
...
7 15 4 21.0526
8 16 1 5.2632 slide 19

Notes: Each row is from which variable?

Add a variable name ’var’ and retain the name of the variable.polishing data
structure

1 PROC freq data = &in_data;
2 tables &var / noprint
3 out = out_freq
4 (rename =(&var = valu_&type) );

6 DATA freq_values;
7 attrib var length = $32
8 valu_c length = $32
9 valu_n length = 8;

10 retain var "&var";
11 set &syslast;
12

13 PROC append data = &syslast
14 base = &out_data;
15 run;

slide 20

Notes: 1. The name of the variable, var, is placed first in the data structure. 2. Style guide:
always end routine or subroutine with a run; statement.

polishing test program
1 options source2;
2 %let in_data = sashelp.class;
3 %let out_data = work.out_data;
4 %let proc_name = ’r15-freq.sas’;
5

6 %let var = sex;
7 %let type = c;
8 %include &proc_name;
9

10 %let var = age;
11 %let type = n;
12 %include &proc_name;
13

14 proc print data = &out_data;
15 run;

slide 21

output, 2nd rough
draft

Obs var valu_c valu_n COUNT PERCENT

1 sex F . 9 47.3684
2 sex M . 10 52.6316
3 age 11 2 10.5263
4 age 12 5 26.3158
5 age 13 3 15.7895
6 age 14 4 21.0526
7 age 15 4 21.0526
8 age 16 1 5.2632

slide 22

output, polished
1 PROC print data = &out_data;
2 title3 &out_data;
3 title4 &in_data ;
4 by name notsorted;
5 id name;
6 run;

slide 23

Notes: Consider using ODS for other destinations.

var valu_c valu_n COUNT PERCENT

sex F . 9 47.3684
M . 10 52.6316

age 11 2 10.5263
12 5 26.3158
13 3 15.7895
14 4 21.0526
15 4 21.0526
16 1 5.2632 slide 24

Notes: Information: double-digit percents indicate variable is discrete.

Top Down, Making List of Variable Names

This sql program lists the data structure in the log.echo data structure

1 PROC sql; describe table sashelp.class;
2 describe table dictionary.columns;
3 quit;

r21-sql-describe.log

create table SASHELP.CLASS( label=’Student Data’)
Name char(8),
Sex char(1),
Age num,
Height num,
Weight num slide 25

Notes: We want a data set with this information.

This is the data structure of the sql dictionary columns.sql dictionary
columns

NOTE: SQL table Dictionary.Columns was created like:
libname char(8) label=’Library Name’,
memname char(32) label=’Member Name’,
memtype char(8) label=’Member Type’,
name char(32) label=’Column Name’,
type char(4) label=’Column Type’,
length num label=’Column Length’,
varnum num label=’Column Number in Table’,

slide 26

Notes: Compare to describe view sashelp.vcolumns;

This sql program makes a list of the variable names in a data set.making list of variable
names

1 PROC sql; create table list_variables as
2 select libname, memname, name, type
3 from dictionary.columns
4 where libname eq "%upcase(&libname)"
5 and memname eq "%upcase(&memname)";
6 quit;

slide 27

Notes: 1. Save this program as make-list-vars-sql.sas.
2. Notice the use of SAS naming conventions to promote readability and reuse.

This sql program echos the data structure to the log and prints the data set.subroutine for testing

1 PROC sql; describe table &syslast;
2 quit;
3 PROC print data = &syslast;
4 title3 &syslast;
5 run;

slide 28

Notes: 1. Programmers want to read the notes in the log produced by sql; users want to
examine the output from the print procedure. 2. This is from r12-freq on page 5.

testing subroutine
make-list 1 %let libname = sashelp;

2 %let memname = class;
3 %include ’r22-sql-make-list.sas’;
4 %include ’describe.sas’;

slide 29

a control data set
Obs libname memname name type

1 SASHELP CLASS Name char
2 SASHELP CLASS Sex char
3 SASHELP CLASS Age num
4 SASHELP CLASS Height num
5 SASHELP CLASS Weight num

slide 30

Processing the List

demo.1 of sql text Notes: The first few lines of program r31-sql-proc-text-demo.sas are not shown on
the slide.

1 %let libname = sashelp;
2 %let memname = class;
3 %include ’make-list-vars-sql.sas’;

5 PROC sql; select ’prefix’, name, ’etc’, type, ’suffix’
6 from list_variables;
7 quit;

r31-sql-proc-text-demo.lst

Name Type
----------------------------------------
prefix Name etc char suffix
prefix Sex etc char suffix
prefix Age etc num suffix
prefix Height etc num suffix
prefix Weight etc num suffix

slide 31

demo.2 of sql text
5 PROC sql; select ’%let name=’, name, ’;’
6 ,’%let type=’, type, ’;’
7 from list_variables;
8 quit;

r32-sql-proc-text-let.lst

Name Type
--------------------------------------------
%let name= Name ;%let type= char ;
%let name= Sex ;%let type= char ;
%let name= Age ;%let type= num ;
%let name= Height ;%let type= num ;
%let name= Weight ;%let type= num ;

slide 32

Notes: 1. Test program, r33-sql-proc-list-test, is not shown in the slides.
2. Program r22-sql-make-list on page 11 is the predecessor of make-list-vars-sql.

calling routine
1 %let libname = sashelp;
2 %let memname = class;
3 %include ’make-list-vars-sql.sas’;
4

5 %let proc_name = ’put-name-type.sas’;
6 %include ’r33-sql-proc-list.sas’;

Notes: This test program, put-name-type.sas, is not shown in the slides.

1 %put note: &=name &=type;

called subroutine
1 PROC sql; select catt(’%let name=’, name, ’;’
2 ,’%let type=’, type, ’;’
3 ,"%include &proc_name ;")
4 into :proc_list separated by ’ ’
5 from list_variables;
6 quit;
7 &proc_list

r33-sql-proc-list.lst

%let name=Name ;%let type=char;%include ’put-name-type.sas’;
%let name=Sex ;%let type=char;%include ’put-name-type.sas’;
%let name=Age ;%let type=num ;%include ’put-name-type.sas’;
%let name=Height;%let type=num ;%include ’put-name-type.sas’;
%let name=Weight;%let type=num ;%include ’put-name-type.sas’;

slide 33

Notes: Save r33-sql-proc-list-test.sas as proc-list-vars.sas,
as shown on page 18.

This program has four parameters, libname, memname, name and type.using SAS names

1 PROC freq data = &libname..&memname;
2 tables &name / noprint
3 out = out_freq
4 (rename =(&name = valu_&type) );
5

6 DATA freq_values;
7 attrib name length = $32
8 valu_char length = $32
9 valu_num length = 8;

10 retain name "&name";
slide 34

Notes: Naming conventions: programs for the frequency procedure have the prefix r1.

first review
1 %let in_data = sashelp.class;
2 %let libname = %scan(&in_data, 1,.);
3 %let memname = %scan(&in_data,-1,.);
4 %let out_data = work.out_data;
5

6 %include ’make-list-vars-sql.sas’;
7

8 %let proc_name = ’r16-freq.sas’;
9 %include ’r33-sql-proc-list.sas’;

10

11 %include ’proc-print-out-data.sas’;
slide 35

Notes: Compare to proc-1-freq.sas, shown on page 18.

valu_
name char valu_num COUNT PERCENT

Name Alfred . 1 5.2632
Alice . 1 5.2632

...
Thomas . 1 5.2632
William . 1 5.2632

Sex F . 9 47.3684
M . 10 52.6316

slide 36

Notes: For variable Name count equals 1 for all values; this means it is unique, the row-
identifier.

Age 11.0 2 10.5263
12.0 5 26.3158

...
15.0 4 21.0526
16.0 1 5.2632

Height 51.3 1 5.2632
...

62.5 2 10.5263
...

66.5 2 10.5263
72.0 1 5.2632

slide 37
Notes: 1. For classification variable Age, n-levels = 6, this means it is discrete.

2. For fact variable Height, many instances of count=1, this means it is continuous.

Change Notice, Adjustments for Summary Procedure

requirements, v2 from:
to:

for each variable in a data set, perform frequency
for any data set

1. character variables, frequency

2. numeric variables, summary

module routine subroutine
02-data-*
proc-2-freq-smry make-list-vars-sql

proc-list-vars-subset
proc-freq data-structure
proc-summary

Notes: Compare to first requirements on page 2.

r51-smry.sas is not shown in the slides.proc summary

1 PROC summary data = sashelp.class;
2 var height;
3 output
4 out = out_summary;
5 %include ’describe.sas’;

Notes: Style guide: align keywords, equal sign, and arguments. Readability promotes reuse.

step.2 soft-coded
1 %let libname = sashelp;
2 %let memname = class;
3 %let name = height;
4

5 PROC summary data = &libname..&memname;
6 var &name;
7 output
8 out = out_summary;
9 %include ’describe.sas’;

slide 39

data structure
create table work.out_summary

(label=’Summary Statistics’)
_TYPE_ num,
_FREQ_ num,
_STAT_ char(8),
Height num r52-smry-test.lst

Obs _TYPE_ _FREQ_ _STAT_ Height

1 0 19 N 19.0000
2 0 19 MIN 51.3000
3 0 19 MAX 72.0000
4 0 19 MEAN 62.3368
5 0 19 STD 5.1271

slide 40

Notes: r53-smry.sas is not shown in the slides.

1 PROC summary data = &libname..&memname;
2 var &name;
3 output
4 out = out_summary
5 ( drop = _type_ _freq_
6 rename = (_stat_ = valu_char
7 &name = valu_num));
8 %include ’describe.sas’;

Note: Style, naming conventions: the suffix of the variables named valu-* is in (’char’,’num’),
which is the value of the variable type in the control data set.

proc summary
1 PROC summary data = &libname..&memname;
2 var &name;
3 output
4 out = out_summary
5 ( drop = _type_ _freq_
6 rename = (_stat_ = valu_char
7 &name = valu_num));
8 %include ’data-structure.sas’;

slide 41

Notes: Save r54-smry.sas as proc-summary.sas; see listing on page 20.

common subroutine
1 DATA standardized_data_structure
2 (label = "freq/smry of &libname &memname");
3 attrib name length = $32
4 valu_char length = $32 %* fragile! ;
5 valu_num length = 8
6 count length = 8
7 percent length = 8;
8 retain name "&name";
9 set &syslast;

10

11 PROC append data = &syslast
12 base = &out_data;
13 run;

slide 42
Notes: See r16-freq.sas on page 13. See listing on page 20.

proc freq, updated
1 PROC freq data = &libname..&memname;
2 tables &name / noprint
3 out = out_freq
4 (rename =(&name = valu_&type));
5

6 %include ’data-structure.sas’;
slide 43

Notes: Save r17-freq.sas as proc-freq.sas, see listing on page 20.

Process the Subset

processing a subset
1 PROC sql noprint;
2 select catt(’%let name=’, name, ’;’
3 ,’%let type=’, type, ’;’
4 ,"%include &proc_name ;")
5 into :proc_list separated by ’ ’
6 from list_variables
7 where &proc_where; * <---<<< ;
8 quit;
9 &proc_list

slide 44
Notes: Save r34-sql-proc-list-where.sas as proc-list-vars-subset.sas,

see listing on page 21.

main module
8 %let proc_name = ’proc-freq.sas’;
9 %let proc_where = type eq ’char’;

10 %include ’proc-list-vars-subset.sas’;
11

12 %let proc_name = ’proc-summary.sas’;
13 %let proc_where = type eq ’num’;
14 %include ’proc-list-vars-subset.sas’;
15

16 %include ’proc-print-out-data.sas’;
slide 45

Notes: Save r42-proc-all-both.sas as proc-2-freq-smry.sas,
see listing on page 19.

Next-to-Last Review

proc either valu_
name char valu_num count percent

Name Alfred . 1 5.2632
...

William . 1 5.2632

Sex F . 9 47.3684
M . 10 52.6316

Age N 19.000 . .
MIN 11.000 . .
MAX 16.000 . .
MEAN 13.316 . .
STD 1.493 . .

Height N 19.000 . .
MIN 51.300 . .
MAX 72.000 . .
MEAN 62.337 . .
STD 5.127 . .

...

slide 46

future list make list with proc contents
process list with call execute

character add accumulating count, percent
extract pattern, calculate max length

numeric add percentile, calculate n-std Grubbs
examine extremes Dixon, Tukey

variables use cardinality ratio
to identify continuous, discrete, unique

slide 47

Notes: Daily Job Diary (DJD) is a record of tasks, accomplishments, bells and whistles to
add, and other notes-to-self.

Summary

Dividing programs into three types: modules, routines and subroutines, makes developingConclusion
and testing applications easier and faster.

The next step is to convert subroutines into macros, which is the topic of the next chapter.

Recommended
Reading

batch : Walsh [16]

design : Brooks Jr. [2], Mythical Man-Month; Brooks Jr. [3], Design of Design; Fehd [9], Macro
Design Ideas; Knuth [12], Literate Programming; Lidwell et al. [13], Principles of Design

DJD : google: DJD Daily Job Diary and see the pdf from
nickleelectrical.com/wp-content/.../09/DAILY-JOB-DIARY-WEEK.pdf

HIPO : IBM [11], Hierarchical Input Process Output

list processing : Fehd [8], Using Sql Select Into; Fehd and Carpenter [10], List Processing Basics

naming conventions : Martin [14], Clean Code; McConnell [15], Code Complete

SmryEachVar : Fehd [6], Data Review Macro FreqAll; Fehd [5], FreqLibname: A Data Review Routine;
Fehd [7], SmryEachVar: A Data Review Routine

testing : Fehd [4], Writing testing-aware programs; Bentley [1], Software Testing Fundamentals

Ronald J. Fehd mailto:Ron.Fehd.macro.maven@gmail.comContact Information:
http://www.sascommunity.org/wiki/Ronald_J._Fehd

education: B.S. Computer Science, U/Hawaii, 1986
SAS User Group conference attendee since 1989

experience: programmer: 30+ years
author: 40+ SUG papers
sas.community.org wiki: 400+ pages

SAS-L: author: 7,000+ messages to SAS-L since 1997

About the author:

Programs: http://www.sascommunity.org/wiki/

SmryEachVar ApDev Introduction

SAS and all other SAS Institute Inc. product or service names are registered trademarks orTrademarks
trademarks of SAS Institute Inc. in the USA and other countries. R© indicates USA registration.
Other brand and product names are trademarks of their respective companies.

Program Listings, Summarize-Each-Variable Suite

Suite.1 Frequency of All Variables

module routine subroutine lines
01-data-* 5
proc-1-freq 16

make-list-vars-sql 6
proc-list-vars 10
proc-print-out-data 7
proc-freq 9

01-data-sashelp-class.sas
1 %let in_data = sashelp.class;
2 %let out_data = work.summary_each_var_class;
3

4 %include ’proc-1-freq.sas’;

proc-1-freq.sas
1 ******** echo included statements to log?;
2 options source2;
3

4 **** split two-level name: in_data :: libref.data;
5 %let libname = %scan(&in_data, 1,.);
6 %let memname = %scan(&in_data,-1,.);
7 %put note: &=in_data &=libname &=memname;
8

9 ******** create input list: control data set;
10 %include ’make-list-vars-sql.sas’;
11

12 ******** call item processing routine;
13 %let proc_name = ’r16-freq.sas’;
14 %include ’proc-list-vars.sas’;
15

16 %include ’proc-print-out-data.sas’;

make-list-vars-sql.sas
1 PROC sql; create table list_variables as
2 select libname, memname, name, type
3 from dictionary.columns
4 where libname eq "%upcase(&libname)"
5 and memname eq "%upcase(&memname)";
6 quit;

proc-list-vars.sas
1 %put note: &=proc_name;
2 *fragile: max length sql cat* values=$char200.;
3 *fragile: max length mvar = 2**16 = 64K;
4

5 PROC sql noprint;
6 select catt(’%let name=’, name, ’;’
7 ,’%let type=’, type, ’;’
8 ,"%include &proc_name ;" )
9 into :process_list separated by ’ ’

10 from list_variables;
11 quit;
12 &process_list

18

Note: proc-freq for suite.1 is r16-freq, on page 13.

proc-print-out-data.sas
1 PROC print data = &out_data;
2 title3 &out_data;
3 title4 &in_data ;
4 by name notsorted;
5 id name;
6 run;

Suite.2 Frequency or Summary

module routine subroutine lines
01-data-* 5
proc-2-freq-smry 22

make-list-vars-sql 6
proc-list-vars-subset 11
proc-print-out-data 7

data-structure 13
proc-freq 9
proc-summary 11

02-data-sashelp-heart.sas
1 %let in_data = sashelp.heart;
2 %let out_data = work.summary_each_var_heart;
3

4 %include ’proc-2-freq-smry.sas’;

proc-2-freq-smry.sas
1 ******** echo included statements to log?;
2 *options source2;
3

4 **** split two-level in-data :: libref.data;
5 %let libname = %scan(&in_data, 1,.);
6 %let memname = %scan(&in_data,-1,.);
7 %put note: &=in_data &=libname &=memname;
8

9 ******** create input list: control data set;
10 %include ’make-list-vars-sql.sas’;
11

12 ******** call item processing routines;
13 %let proc_name = ’proc-freq.sas’;
14 %let proc_where = 1; * all? or only char?;
15 %let proc_where = type eq ’char’;
16 %include ’proc-list-vars-subset.sas’;
17

18 %let proc_name = ’proc-summary.sas’;
19 %let proc_where = type eq ’num’;
20 %include ’proc-list-vars-subset.sas’;
21

22 %include ’proc-print-out-data.sas’;

proc-list-vars-subset.sas
1 %put note: &=proc_name &=proc_where;
2 PROC sql noprint;
3 select catt(’%let name=’, name, ’;’

4 ,’%let type=’, type, ’;’
5 ,"%include &proc_name ;")
6 into :_process_list separated by ’ ’
7 from list_variables
8 where &proc_where; * <---<<< ;
9 quit;

10 &_process_list

data-structure.sas
1 DATA standardized_data_structure
2 (label = "freq/smry of &libname &memname");
3 attrib name length = $32
4 valu_char length = $32 %* fragile! ;
5 valu_num length = 8
6 count length = 8
7 percent length = 8;
8 retain name "&name";
9 set &syslast;

10

11 PROC append data = &syslast
12 base = &out_data;
13 run;

proc-freq.sas
1 %put note: &=libname &=memname &=name &=type;
2

3 PROC freq data = &libname .&memname;
4 tables &name / noprint
5 out = out_freq
6 (rename =(&name = valu_&type));
7

8 %include ’data-structure.sas’;

proc-summary.sas
1 %put note: &=libname &=memname &=name &=type;
2

3 PROC summary data = &libname..&memname;
4 var &name;
5 output
6 out = out_summary
7 ( drop = _type_ _freq_
8 rename = (_stat_ = valu_char
9 &name = valu_num));

10

11 %include ’data-structure.sas’;

Program Listings, testing programs

routine subroutine lines
make-list-vars-sql-test 6

proc-list-vars-subset-test 14
proc-freq-test 9
proc-summary-test 6

describe 7

make-list-vars-sql-test.sas
1 %let libname = sashelp;
2 %let memname = class;
3 %include ’make-list-vars-sql.sas’;
4 %include ’describe.sas’;

proc-freq-test.sas
1 %let libname = sashelp;
2 %let memname = class;
3 %let out_data = out_data_&memname;
4

5 %let name = sex;
6 %let type = char;
7 %include ’proc-freq.sas’;
8 %include ’describe.sas’;
9

10 %let name = age;
11 %let type = num;
12 %include ’proc-freq.sas’;
13 %include ’describe.sas’;

proc-list-vars-subset-test.sas
1 %let libname = sashelp;
2 %let memname = class;
3 %let out_data = work.summary_each_var_class;
4

5 %include ’make-list-vars-sql.sas’;
6 %let proc_name = ’put-name-type.sas’;
7 %let proc_where = 1;
8 %include ’proc-list-vars-subset.sas’;

proc-summary-test.sas
1 %let libname = sashelp;
2 %let memname = class;
3 %let out_data = out_data_&memname;
4

5 %let name = height;
6 %let type = num;
7 %include ’proc-summary.sas’;
8 %include ’describe.sas’;
9

10 %let name = weight;
11 %include ’proc-summary.sas’;
12 %include ’describe.sas’;

