
ABSTRACT

SAS® provides a wealth of resources for users to create useful, attractive metadata tables, including PROC
CONTENTS listing output (to ODS destinations), the PROC CONTENTS OUT= SAS data set, and PROC
CONTENTS ODS Output Objects. This paper and presentation explore some less well-known resources to create
metadata such as %SYSFUNC, PROC DATASETS, and Dictionary Tables. All these options in conjunction with the
use of the EXCELXP (and new in 9.4 M2, the EXCEL) tagsets enable the creation of multi-tab metadata workbooks
at the click of a mouse.

INTRODUCTION

In a prior paper (Hadden, 2014) I discuss using PROC CONTENTS, ODS OUTPUT and the EXCELXP tagset to
create multi-tab documentation (metadata) spreadsheets to describe data sets. This paper and presentation build on
that concept, exploring several alternative procedures and tools to view and create SAS metadata. Although all the
tools discussed have utility beyond documentation purposes, other uses are beyond the scope of this paper. Users
are encouraged to read the papers referred to in the References section for more information, and to explore these
great resources on their own.

Who cares about metadata? Any SAS programmer should care! Knowing your data is vital. By using SAS metadata
in conjunction with careful documentation practices that some of the tools we are discussing make easy, you can find
out when a program was last run, who ran it, what variables were created, whether the data set is sorted or indexed,
and more. Follow me down the metadata worm hole and we will find out how….

RESOURCES WITHIN SAS® TO OBTAIN METADATA

Every time a SAS session is initiated, a wealth of metadata becomes available to users. The SAS system collects
information on libraries and data sets specified in LIBNAME and FILENAME statements, libraries and data sets
automatically available to SAS users, data sets and files created within the session, system and version information,
host specific information such as usernames, among others. This paper discusses five methods of accessing this
information (PROC CONTENTS having been covered earlier, making six methods.)

1) PROC CONTENTS (not covered here)

2) PROC DATASETS

3) SAS Dictionary Tables (used with PROC SQL)

4) SASHELP Views (can be used with PROC SQL and with the SAS data step)

5) %SYSFUNC (access to metadata through MACRO functions)

6) “V” functions (access to metadata through data set functions)

I will discuss resources 2 through 6, briefly describing the resource in relation to metadata, and showing an example
of using each resource. I hope the utility of each resource, whichever one(s) appeal to a given SAS programmer, will
be obvious.

RESOURCE #2. PROC DATASETS

PROC DATASETS does all that PROC CONTENTS does, and much, much more. The primary use I will discuss is
the CONTENTS listing, although we will touch on PROC DATASETS ability to modify datasets by adding labels,
formats, etc. An advantage that PROC DATASETS has over PROC CONTENTS is that it can produce a single file
containing documentation on multiple data sets in a library with a single command.

* RESOURCE 2: PROC DATASETS;

/* generate a contents listing of all data sets in SASHELP */
PROC DATASETS LIBRARY=SASHELP NOLIST;

CONTENTS DATA=_ALL_
OUT=LIB.SASHELP_DS_METADATA (KEEP=LIBNAME MEMNAME NAME TYPE LENGTH LABEL

FORMAT NOBS COMPRESS SORTED IDXUSAGE CRDATE MODATE)
NOPRINT DIRECTORY;

RUN;

/* Add a data set label */
PROC DATASETS LIBRARY=LIB;

MODIFY SASHELP_DS_METADATA
(LABEL="Metadata for all SAS data sets in SASHELP");
CONTENTS DATA=SASHELP_DS_METADATA;

RUN:
QUIT;

A contents listing and a test print of the resulting data set (LIB.SASHELP_DS_METADATA) is listed in Outputs 1
and 2 respectively. Note that some variables in the output data set are on the data set level, while others are on the
variable level, just as in PROC CONTENTS. ODS OUTPUT objects from PROC DATASETS output can be
manipulated just as they can be for PROC CONTENTS, and the output data set can be manipulated with data step(s)
as well.

Output 1. Contents listing output from PROC DATASETS

Output 2. Test print of output data set created by PROC DATASETS

RESOURCE #3. SAS DICTIONARY TABLES

There are many excellent papers on SAS DICTIONARY TABLES and PROC SQL and I urge SAS users to take
advantage of the very useful information in them. Like PROC DATASETS, the utility of SAS DICTIONARY TABLES
goes way beyond documentation and metadata purposes. In example 3, I demonstrate how to use dictionary tables
to find all data sets in a library containing a variable named ZIP. Dictionary tables are accessed using PROC SQL.

* RESOURCE 3: SAS DICTIONARY TABLES;
/* select data sets in a library with the zip variable */
PROC SQL;

TITLE1 'ALL DATA SETS IN SASHELP WITH A VARIABLE NAMED ZIP';
SELECT LIBNAME, MEMNAME, NAME FROM DICTIONARY.COLUMNS
WHERE NAME='ZIP' and libname='SASHELP';

QUIT;

The listing from PROC SQL is shown in Output 3. We can see that three data sets in the SASHELP library have a
variable named ZIP. We could have selected additional variables such as type, index information, and sort
information to ensure any merges would be successful, and we could have created a permanent data set to do
further manipulations, as SAS Dictionary Tables “disappear” when a SAS session ends.

Output 3. Listing of SAS Dictionary Information produced by PROC SQL

RESOURCE #4. SASHELP VIEWS

SASHELP views are analogous to dictionary tables. They are SAS views based on the same metadata that the
dictionary tables are based on. They may be accessed using PROC SQL or with data step(s).

* RESOURCE 4: SASHELP VIEWS;
PROC SQL;

CREATE TABLE LIB.SASHELP_ZIPCODE_CDBK AS
SELECT NAME, TYPE, LENGTH, FORMAT, VARNUM, IDXUSAGE, SORTEDBY, LABEL
FROM SASHELP.VCOLUMN

WHERE LIBNAME='SASHELP' AND MEMNAME='ZIPCODE';
QUIT;

PROC PRINT DATA=LIB.SASHELP_ZIPCODE_CDBK (OBS=10) NOOBS;
RUN;

The listing from PROC SQL is shown in Output 4. We could have used a data step in this case, but PROC SQL is
much more efficient. We have created a permanent data set as the basis of a data dictionary, as the SAS view will
not be accessible once our SAS session has ended.

Output 4. Listing of SASHELP View Information produced by PROC SQL

RESOURCE #5. %SYSFUNC

%SYSFUNC (or %QSYSFUNC) is a macro function that allows you to access SAS language functions regardless of
the system environment you are operating in. For example, %IF %SYSFUNC(EXIST(infi)) %THEN %DO; . . . %END;
sets up conditional processing based on infi’s existence. Selected SAS® functions that can be used with
%SYSFUNC: Verify existence of a file; open a file; get information about a file, delete a file, close a file, read a host-
specific option. Unlike PROC CONTENTS, PROC DATASETS, Dictionary Tables, and SASHELP views,
%SYSFUNC does produce tables and views, but instead uses macro processing to obtain SAS metadata and turn
selected items into macro variables. These macro variables can be used in a number of ways. They can be used to
create variables that can be used for documentation, to support conditional processing based on whether a file exists
or not or whether a variable is character or numeric, to add documentation details to data set labels, variable labels,
and catalog descriptions, what user name last run a program, etc. It is a very powerful tool and can be used in %LET
statements, in titles/footnotes/ODS text/descriptions as macro variables, and incorporated into macro routines.
Consider the following title statement:

TITLE1 " Created on &SYSDATE - &SYSTIME - by &SYSUSERID - using
%SYSFUNC(GETOPTION(SYSIN))";

This single statement provides information to both the log and list as to when the program was last run, who ran it,
and what the program name is. Similar text can be used to describe the data set in a data set label, describe
variables, and so on.

The example shown is to write the number of observations in a data set to the log. Why bother? We perform
monthly processing of data, and just once (last month) a data set was created with 0 observations, an error which did
not trigger an ERROR statement in the log and had a domino effect on our processing routine. Early warning with a
few %LET statements written to the log is a small price to pay!

* RESOURCE 5: %SYSFUNC;
/* find the number of observations in a data set */

/* check to see that the data set exists */

%LET dsexist=%SYSFUNC(EXIST(sashelp.zipcode));
%PUT &dsexist;

/* identify the data set */

%LET sasdsname=sashelp.zipcode;

/* find the data set's id */

%LET dsid=%SYSFUNC(OPEN(&sasdsname));

%put DSID = &dsid;

/* find the attribute # of obs from data set with dsid */

%LET numobs=%SYSFUNC(ATTRN(&dsid,NOBS));

/* close the data set with dsid */

%LET rc=%SYSFUNC(CLOSE(&dsid));

/* put the number of observations in the log */

%PUT The # of observations in &sasdsname is &numobs;

A snippet from the log is shown in Output 5.

Output 5. Log with data set level information produced by %SYSFUNC

RESOURCE #6. “V” FUNCTIONS

“V” functions are analogous to %SYSFUNC macro functions, but are less powerful, specific to variables and used
within the SAS data step. Like %SYSFUNC, it is a good resource for enabling conditional processing based on
variable type, formats, etc.

* RESOURCE 6: “V” FUNCTIONS;
DATA dateformattest;

LENGTH srvy_dt_type . . . $ 1
srvy_dt_fmt . . .$ 32;

SET defs.alltexts_indx20150301;
IF _n_=1;

srvy_dt_type=VTYPE(srvy_dt); . . .
srvy_dt_fmt=VFORMAT(srvy_dt); . . .

RUN;

The listing from PROC PRINT of the variables created using a data step and “V” functions is shown in Output 6.
This could prevent one of those “gotchas” where dates come in as numeric one month and character another, but
with the same variable names, or come in as DATETIME20. one month and MMDDYY10. another month.

Output 6. Print out of date variable types and formats produced using “V” functions

CONCLUSION

If you are looking to find SAS metadata, there’s more than one way to skin a cat! Take advantage of the master
procedure PROC DATASETS (as well as PROC CONTENTS), SAS Dictionary Tables and SASHELP.VIEWS, “V”
functions and information gleaned from using the powerful %SYSFUNC to create user documentation, generate SAS
code, perform conditional processing and GET INPUT!

