
ABSTRACT 

Normally people would think titles and/or footnotes are shown, by default, in the ODS statistical graphics procedures 
(SG procedures), such as PROC SGPLOT, provided that they are not turned off.  However, when using these 
procedures with the ODS PDF statement, this is surprisingly untrue.  Interestingly, by adding ODS LAYOUT 
statement, magically those titles or footnotes pop out in the output!  Also, placing TITLE or FOOTNOTE statements 
before or after ODS LAYOUT shows different results.  This presentation will use examples to demonstrate the results. 

INTRODUCTION 

In a statistical analysis and reporting environment, a clear presentation of analysis data and graphs is essential.  Not 
only are the data themselves informative, but also, proper titles and/or footnotes help clearly communicate the results 
of the analysis to data consumers.  Fortunately, SAS provides a number of tools to present results; one of the most 
popular tools is ODS, due to its functionality. 

ODS stands for Output Delivery System, which produces SAS procedures and DATA step output with a wide range of 
formatting options that can output to many different destinations, such as HTML, RTF, PDF, and so on.  For example, 
if you choose PDF as the destination, then by using the ODS PDF statement, it produces a PDF output. 

Other than providing formatting options and different destinations, ODS has included a list of statistical graphics 
procedures (SG procedures): 

SGPLOT 

SGPANEL 

SGSCATTER 

SGRENDER 

SGDESIGN 

These graphics procedures use ODS destination statements to control where the output goes and how it looks. 

In most SAS environments, employing the TITLE or FOOTNOTE statements will generate titles or footnotes in the 
results.  However, this is untrue when using ODS PDF to display ODS statistical graphics procedures results, despite 
using the statements above, one is unable to generate the titles and footnotes. 

How do we fix this? 

The magic statement is ODS LAYOUT.  The ODS LAYOUT statement is in preproduction and experimental for SAS 
9.0 through SAS 9.3.  However, it is in production starting with SAS 9.4.  This paper will demonstrate how using ODS 
PDF and PROC SGPLOT with ODS LAYOUT, produces graphs with titles and footnotes.  The examples below are 
based on UNIX SAS 9.4. 

EXAMPLES 

The following examples use ODS PDF to output graphs using PROC SGPLOT with a title and a footnote.  We will 
show what happens to the title and the footnote in the following scenarios: 

1. Single Page Output 

a. Without ODS LAYOUT 

b. With ODS LAYOUT positioned in a particular order within the code 

2. Multiple Page Output 

a. With ODS LAYOUT positioned in a particular order within the code 

EXAMPLE 1A – WITHOUT ODS LAYOUT 

Result – No matter where the TITLE and FOOTNOTE statements are located in the code, no title or footnote will be 
seen in the output. 

Code Output 

/* no title and footnote will be in the output file */ 

 

ods listing close; 

ods pdf file="&output_loc./example1a.pdf"; 

 

title; 

footnote; 

title1 "title1"; 

footnote1 "footnote1"; 

 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=actual 

               barwidth=0.6; 

run; 

 

ods pdf close; 

ods listing; 

 

Table 1.  Without ODS LAYOUT 

EXAMPLE 1B – WITH ODS LAYOUT PLACED AFTER TITLE AND FOOTNOTE 

Result – the title and the footnote are shown. 

Code Output 

/* title and footnote will be in the output file */ 

 

ods listing close; 

ods pdf file="&output_loc./example1b.pdf"; 

 

title; 

footnote; 

title1 "title1"; 

footnote1 "footnote1"; 

 

ods layout start; 

 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=actual 

               barwidth=0.6; 

run; 

 

ods layout end; 

 

ods pdf close; 

ods listing; 
 

Table 2.  With ODS LAYOUT Placed After TITLE and FOOTNOTE 

ODS LAYOUT should be placed after TITLE and FOOTNOTE in order to display them in the PDF file.  If they are 
placed in any other positions in the code, such as Position 1 in the Table 2 sample code, the title and the footnote will 
not be seen in the PDF output file. 

From Example 1a and 1b, we can summarize that in order to display titles and/or footnotes to the ODS PDF 
destination with SG procedures, ODS LAYOUT has to be used after the TITLE/FOOTNOTE statements. 

EXAMPLE 2A – WITH ODS LAYOUT POSITIONED IN A PARTICULAR ORDER WITHIN THE CODE FOR A MULTIPLE PAGE 
OUTPUT. 

Let’s look at several scenarios where TITLE and FOOTNOTE statements are in different places in the code.  Of note: 
we added the STARTPAGE option to control outputting graphs to multiple pages. 

Position1 



Where do the titles or footnotes go when using PROC SGPLOT in ODS PDF?, continued SESUG 2015 
 

3 

Scenario A – Follow the rule defined in Example1b where ODS LAYOUT is placed after TITLE and FOOTNOTE 
for each page. 

Result –Titles and footnotes will be appropriately displayed on each page. 

Code Output 

ods listing close; 

ods pdf file="&output_loc./example2a.pdf" 

startpage=never; 

 

/* first page title and footnote */ 

title; 

footnote; 

title1 "title1"; 

footnote1 "footnote1"; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* the first sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=actual 

               barwidth=0.6; 

run; 

 

ods layout end; 

 

/* second page title and footnote */ 

title; 

footnote; 

title1 "title2"; 

footnote1 "footnote2"; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* the second sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=predict 

               barwidth=0.3; 

run; 

 

ods layout end; 

 

ods pdf close; 

ods listing; 

Page1 

 

 

 

Page 2 

 

 

 

Table 3.  ODS LAYOUT is placed after TITLE and FOOTNOTE for each page 

Scenario B – ODS LAYOUT is placed after the TITLE and FOOTNOTE in the first page, however on the 
second page, ODS LAYOUT is placed before TITLE and FOOTNOTE. 

Result – The title and footnote are not seen at all on the second page.  They are only seen on the first page where we 
placed ODS LAYOUT after TITLE and FOOTNOTE.  As such, this reinforces that the placement of ODS LAYOUT is 
important. 

Code Output 

ods listing close; 

ods pdf file="&output_loc./example2b.pdf" 

startpage=never; 

 

/* first page title and footnote */ 

title; 

footnote; 

title1 "title1"; 

footnote1 "footnote1"; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* the first sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=actual 

               barwidth=0.6; 

run; 

 

ods layout end; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* second page title and footnote */ 

title; 

footnote; 

title1 "title2"; 

footnote1 "footnote2"; 

 

/* the second sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=predict 

               barwidth=0.3; 

run; 

 

ods layout end; 

 

ods pdf close; 

ods listing; 

Page1 

 

 

 

Page 2 

 

 

 

Table 4.  ODS LAYOUT is placed after the TITLE and FOOTNOTE in the first page and before the TITLE and 
FOOTNOTE in the second page 

Scenario C – ODS LAYOUT is placed before the TITLE and FOOTNOTE in the first and second pages. 

Result – The title and footnote do not display on the first page, instead, the first page’s title and footnote show on the 
second page.  This is not what is expected. 

Code Output 

ods listing close; 

ods pdf file="&output_loc./example2c.pdf" 

startpage=never; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* first page title and footnote */ 

title; 

footnote; 

title1 "title1"; 

footnote1 "footnote1"; 

 

/* the first sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=actual 

               barwidth=0.6; 

run; 

 

ods layout end; 

 

ods pdf startpage=now; 

ods layout start; 

 

/* second page title and footnote */ 

title; 

footnote; 

title1 "title2"; 

footnote1 "footnote2"; 

 

/* the second sgplot */ 

proc sgplot data=sashelp.prdsale; 

   vbar year / response=predict 

               barwidth=0.3; 

run; 

 

ods layout end; 

 

ods pdf close; 

ods listing; 

Page1 
 

 
 
Page 2 
 

 
 

Table 5.  ODS LAYOUT is placed before TITLE and FOOTNOTE for each page 

CONCLUSION 

Contrary to people’s perceptions that titles and/or footnotes are simple and easy to use, the usage in ODS PDF with 
SG procedure outputs is not straightforward.  Fortunately with ODS LAYOUT placed in a particular position in the 
code, titles and/or footnotes are able to be displayed properly. 

